# Difference Equations

Difference equation is a recurrence relation between the terms of some sequence. Let ( $a_n$ ) be a real numbers sequence.
First oder difference equation: $a_{n+1} = \mathrm{f}(a_n)$
Second order difference equation: $a_{n+2} = \mathrm{f}(a_n, a_{n+1})$
k-order difference equation: $a_{n+k} = f(a_n, a_{n+1}, \ldots, a_{n+k-1})$
Difference equations generate dynamical systems.

## Long term behaviour of $a_{n+1} = r\, a_n$ for $r$ constant

For the linear difference equation $a_{n+1} = r\, a_n$ , let's consider some significant values of $r$ .
If $r = 0$ then $a_n = 0$ (except possibly a[0]) for all n>0, so there is no need for further investigation.
If $r = 1$ then $a_{n+1} = a_n = a_0$ , in this case we have a constant sequence.

What happens for some other values?

Let's cosider $a_{n+1} = r\, a_n$ , for $r = .5$ and $a_0 = 3$

```
> a[0]:=3;r:=0.5;
```

$$a_0 := 3$$

$$r := 0.5$$

```
> for i from 0 to 9 do
  a[i+1]:=r*a[i]
end do:
> plot([[n,a[n]]$n=0..10],style=point,symbol=circle);
```




For this example we notice that the terms of the sequence are decreasing and it seems that $a_n$ tends to 0.

Let's cosider $a_{n+1} = r\,a_n$, for $r = 1.02$ and $a_0 = 3$

```
> a[0]:=3;r:=1.02;
> for i from 0 to 19 do
   a[i+1]:=r*a[i]
end do:
```

$$a_0 := 3$$

$$r := 1.02$$

```
> plot([[n,a[n]]$n=0..20],style=point,symbol=circle);
```




For this example we notice that the terms of the sequence are increasing unbounded.

Let's cosider $a_{n+1} = r\,a_n$, for $r = -.8$ and $a_0 = 3$

```
> a[0]:=3;r:=-0.8;
> for i from 0 to 19 do
   a[i+1]:=r*a[i]
end do:
> plot([[n,a[n]]$n=0..20],style=point,symbol=circle);
```

$$a_0 := 3$$

$$r := -0.8$$



In this case the terms of the sequence oscilate and it seems that $a_n$ tends to 0.

Let's cosider $a_{n+1} = r\, a_n$ , for $r = -1.08$ and $a_0 = 3$

```
> a[0]:=3;r:=-1.08;
```

$$a_0 := 3$$

$$r := -1.08$$

```
> for i from 0 to 19 do
   a[i+1]:=r*a[i]
end do:
> plot([[n,a[n]]$n=0..20],style=point,symbol=circle);
```



In this case the terms of the sequence oscilate and it seems that the oscilations growth.

In fact, if we start with some initial data $a_0$ we have:

$a_1 = r\,a_0$

$a_2 = r\,a_1 = r^2\,a_0$

...........

$a_n = r\,a_{n-1} = r^n\,a_0$

Therefore, the behaviour is given by the value of $r$ . The expression $a_n = r^n\,a_0$ represent the solution of the difference equation. Thus, we have:

I. if $|r| < 1$ then $a_n$ tends to 0 for all initial data $a_0$

II. if $r > 1$ then $a_n$ tends to $\mathrm{sign}(a_0)\,\infty$ without oscillations

III. if $r < -1$ then $|a_n|$ tends to + $\infty$ and the terms of the sequence alternate the sign.

## Long term behaviour of $a_{n+1} = r\,a_n + b$ , where $r$ and $b$ are constants

**Exercise**: *Test the following cases:*

$a_{n+1} = .5\,a_n + .1$ , for $a_0 = .1$ , $a_0 = .2$ and $a_0 = .3$

$a_{n+1} = 1.01\,a_n - 1000$ , for $a_0 = 90000$ , $a_0 = 100000$ and $a_0 = 110000$

If $r = 0$ then $a_n = b$ for all n > 0, so we have a constant sequence (excepting $a_0$ )

If $r = 1$ then $a_{n+1} - a_n = b$ for all n, so

$a_1 - a_0 = b$

$a_2 - a_1 = b$

..................

$a_{n-1} - a_{n-2} = b$

$a_n - a_{n-1} = b$

-------------------

$a_n - a_0 = n\,b$

The solution is: $a_n = n\,b + a_0$ , the terms of the solution are on the line $y = b\,x + a_0$

If $r \neq 1$ and $r \neq 0$, then

$a_1 = r\,a_0 + b$

$a_2 = r\,a_1 + b = r\,(r\,a_0 + b) + b = r^2\,a_0 + b\,(r + 1)$

$a_3 = r\,a_2 + b = r\,(r^2\,a_0 + b\,(r + 1)) = r^3\,a_0 + b\,(r^2 + r + 1)$

.............

$$a_n = r^n\,a_0 + b\,(r^{(n-1)} + ... + r + 1) = r^n\,a_0 + \frac{b\,(1 - r^n)}{1 - r} = r^n\,a_0 + \frac{b}{1 - r} - \frac{b\,r^n}{1 - r}$$

The solution is $a_n = r^n \left( a_0 - \frac{b}{1 - r} \right) + \frac{b}{1 - r}$

**Equilibrium solution. Equilibrium point.**

A constant solution $a_n = a$, for all $n$, of the difference equation $a_{n+1} = \mathrm{f}(a_n)$ is called an *equilibrium solution*.

The value $a$ is called the *equilibrium point* of the difference equation.

Since the equilibrium solution is a constant sequence then the value $a$ is a solution of the equation $a = \mathrm{f}(a)$

For the first order linear difference equation $a_{n+1} = r\,a_n + b$ we have the equilibrium point $a = r\,a + b$

So, $a = \frac{b}{1 - r}$ if $r \neq 1$

If $r = 1$ then the equation has no equilibrium point.

If $r = 0$ and $b = 0$ then equation has an infinite number of equilibrium points (every initial value is an equilibrium point).

In the case of our examples,

$a_{n+1} = .5\,a_n + .1$ the equilibrium point is $a = .2$

$a_{n+1} = 1.01\,a_n - 1000$ the equilibrium point is $a = 100000$

**Stability of an equilibrium point**

**Definition.** The equilibrium point $a$ is *locally stable* equilibrium point if for all solutions ( $a_n$ ) of the difference equation with the initial condition $a_0$ near the value $a$ we have that $a_n \to a$ as $n \to \infty$.

If this property holds for any initial condition $a_0$ then we say that the equilibrium point is *globally stable*.

If the equilibrium point is not locally stable then we say that is *unstable*.

In the case of the first order linear difference equation $a_{n+1} = r\,a_n + b$ the solution is

$$a_n = r^n\left(a_0 - \frac{b}{1-r}\right) + \frac{b}{1-r}$$

so, we have the following situations

I. If $|r| < 1$ then equilibrium point is globally stable

II. If $1 < |r|$ then equilibrium point is unstable

III. If $r = 1$ then the difference equation has no equilibrium point (the graph is a line)

## Nonlinear difference equation $a_{n+1} = \mathrm{f}(a_n)$

Generally, the nonlinear difference equations can not be solved. For these equations it is important to study the behaviour of the solutions with respect to the equilibrium points.

**Exercise**: *Plot the solutions of* $a_{n+1} = r\,(1 - a_n)\,a_n$ *for the following initial data:*

$r = .8$, $a_0 = .5$

$r = 1.5$, $a_0 = .1$

$r = 2.75$, $a_0 = .1$

$r = 3.25$, $a_0 = .1$

$r = 3.525$, $a_0 = .1$

$r = 3.555$, $a_0 = .1$

$r = 3.75$, $a_0 = .1$

```
> a[0]:=0.5;r:=0.8;
```

$$a_0 := 0.5$$

$$r := 0.8$$

```
> for i from 0 to 99 do
  a[i+1]:=r*(1-a[i])*a[i]
end do:
> plot([[n,a[n]]$n=0..100],style=point,symbol=circle);
```




```
> a[0]:=0.1;r:=1.5;
```

$$a_0 := 0.1$$

$$r := 1.5$$

```
> for i from 0 to 99 do
   a[i+1]:=r*(1-a[i])*a[i]
  end do:
> plot([[n,a[n]]$n=0..100],style=point,symbol=circle);
```




```
> r:=2.75;
```

$$r := 2.75$$

```
> for i from 0 to 99 do
   a[i+1]:=r*(1-a[i])*a[i]
  end do:
> plot([[n,a[n]]$n=0..100],style=point,symbol=circle);
```




```
> r:=3.25;
```

$r := 3.25$

```
> for i from 0 to 99 do
   a[i+1]:=r*(1-a[i])*a[i]
  end do:
> plot([[n,a[n]]$n=0..100],style=point,symbol=circle);
```




```
> r:=3.525;
```

$r := 3.525$

```
> for i from 0 to 99 do
   a[i+1]:=r*(1-a[i])*a[i]
  end do:
> plot([[n,a[n]]$n=0..100],style=point,symbol=circle);
```




```
> r:=3.555;
```

$r := 3.555$

```
> for i from 0 to 99 do
   a[i+1]:=r*(1-a[i])*a[i]
end do:
> plot([[n,a[n]]$n=0..100],style=point,symbol=circle);
```




```
> r:=3.75;
```

$r := 3.75$

```
> for i from 0 to 99 do
   a[i+1]:=r*(1-a[i])*a[i]
end do:
> plot([[n,a[n]]$n=0..100],style=point,symbol=circle);
```




**Theorem (Stability in the first approximation)**
Let's consider the difference equation $a_{n+1} = f(a_n)$. Suppose that $a$ is an equilibrium point, i.e. $a = f(a)$, and $f\,'(a)$ is not 0. Then
(i) if $|\,f\,'(a)\,| < 1$ then $a$ is locally stable;
(ii) if $|\,f\,'(a)\,| > 1$ then $a$ is unstable.

**Exercise:** *Using the Theorem of Stability in the first approximation study the difference equation* $a_{n+1} = r\,(1 - a_n)\,a_n$.

**The Stair Step (Cobweb) Diagrams**

Given $a_0$, we pinpoint the value $a_1 = f(a_0)$ by drawing a vertical line through $a_0$ so that it also intersects the graph of $f$ at ( $a_0$, $a_1$ ). Next, draw a horizontal line from ( $a_0$, $a_1$ ) to meet the diagonal line y = x at the point ( $a_1$, $a_1$ ). A vertical line drawn from the point ( $a_1$, $a_1$ ) will meet the graph of $f$ at the point ( $a_1$, $a_2$ ). Continuing this process, one may find $a_n$ for all n > 0.

```
> with(plots):
Warning, the name changecoords has been redefined
```

```
> a[0]:=0.5;r:=0.8;N:=100;
```

$$a_0 := 0.5$$

$$r := 0.8$$

$$N := 100$$

```
> f:=x->r*(1-x)*x;
```

$$f := x \to r\,(1 - x)\,x$$

```
> for i from 0 to N-1 do
  a[i+1]:=f(a[i])
end do:
> plot([[n,a[n]]$n=0..N],style=point,symbol=circle);
```




```
> for i from 1 to N do
  a[i]:=f(a[i-1]):
  lp[2*i-1]:=[a[i-1],a[i-1]];
  lp[2*i]:=[a[i-1],a[i]];
end:
> g1:=plot([lp[j]$j=1..2*N],style=line,color=black):
> g2:=plot([f(x),x],x=0..1):
> display(g1,g2);
```



```
> cobweb:=proc(f,xmin,xmax,a0,n)
 local i,j,x,a,l,g1,g2;
 a[0]:=a0;
 for i from 1 to n do
  a[i]:=f(a[i-1]):
  l[2*i-1]:=[a[i-1],a[i-1]]:
  l[2*i]:=[a[i-1],a[i]]:
end:

g1:=plot([f(x),x],x=xmin..xmax):
g2:=plot([l[j]$j=1..2*n],style=line,color=black):
display(g1,g2);
end:
> cobweb(f,0,1,a[0],N);
```




```
> a[0]:=0.1;r:=1.5;
```

$$a_0 := 0.1$$

$$r := 1.5$$

```
> cobweb(f,0,1,a[0],N);
```




```
> r:=2.75;cobweb(f,0,1,a[0],N);
```

$r := 2.75$



```
> r:=3.25;cobweb(f,0,1,a[0],N);
```

$$r := 3.25$$




```
> r:=3.525;cobweb(f,0,1,a[0],N);
```

$$r := 3.525$$




```
> r:=3.555;cobweb(f,0,1,a[0],N);
```

$$r := 3.555$$



```
> r:=3.75;cobweb(f,0,1,a[0],N);
```

$r := 3.75$



## Periodic Points and Cycles

**Definition.** Let $b$ be in the domain of $f$. Then:
(i) b is called a periodic point of f if for some positive integer $k$, $f^k$ (b) = b. Hence **a point is k-periodic** if it is a fixed point of $f^k$, that
is, if it is an equilibrium point of the difference equation
$a_{n+1} = g(a_n)$, where $g = f^k$.
The periodic orbit of b, O(b) = {b, $f$(b), $f^2$ (b), . . . , $f^{(k-1)}$ (b)}, is often called a **k-cycle**.
(ii) b is called **eventually k-periodic** if for some positive integer $m$, $f^m$ (b) is a k-periodic point. In other words, b is eventually k-periodic if
$f^{(m+k)}$ (b) = $f^m$ (b).

```
> r:='r';
```

$r := r$

```
> f:=x->r*(1-x)*x;
```

$f := x \to r(1-x)x$

```
> g:=x->f(f(x));
```

$g := x \to f(f(x))$

```
> r:=3.25;
```

$r := 3.25$

```
> plot([g(x),x],x=0..1);
```




```
> solve(g(x)=x,x);
```

0., 0.6923076923, 0.8124271394, 0.4952651682

```
> a[0]:=0.1;N:=100;
```

$a_0 := 0.1$

$N := 100$

```
> cobweb(g,0,1,a[0],N);
```




```
> g2:=x->(f@@4)(x);
```

$g2 := f^{(4)}$

```
> r:=3.525;
```

$r := 3.525$

```
> plot([g2(x),x],x=0..1);
```




```
> solve(g2(x)=x,x);
```

0., 0.04878127727 − 0.02298727936 *I*, 0.04878127727 + 0.02298727936 *I*,
0.1654285294 − 0.07312465033 *I*, 0.1654285294 + 0.07312465033 *I*, 0.3709072463,
0.4232189694, 0.5055172382 − 0.1724812237 *I*, 0.5055172382 + 0.1724812237 *I*,
0.5146141230, 0.7163120567, 0.8225060896, 0.8604689738, 0.8804971565,
0.9860106474 − 0.006708920926 *I*, 0.9860106474 + 0.006708920926 *I*

```
> cobweb(g2,0,1,a[0],N);
```




**Theorem.** (Stability of k-cycle)

Let O(b) = $\{b = a_0,\ f(b), f^2(b), \ldots, f^{(k-1)}(b)\}$ be a k-cycle of a continuously differentiable function $f$.

Then the following statements hold:

(i) The k-cycle O(b) is asymptotically stable if

$$|\, f'(a_0)\, f'(a_1) \ldots f'(a_{k-1})| < 1$$

(ii) The k-cycle O(b) is unstable if

$$|\, f'(a_0)\, f'(a_1) \ldots f'(a_{k-1})| > 1$$